Math 277
Ordinary Differential Equations
Fall 2005

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 13

Existence and Uniqueness, Linear Independence
General Solutions, the Wronskian

Theorem 1 (Existence and Uniqueness)

For any choice of the real constants $a \neq 0$, b, c, t_0, Y_0, Y_1 the initial value problem

$$a\,y'' + b\,y' + c\,y = 0 \qquad y(t_0) = Y_0,\ y'(t_0) = Y_1$$

has a unique solution. The solution is valid on $(-\infty, \infty)$.

Note

In all definitions, lemmas, and theorems on this page, the constants $a \neq 0$, b, c, t_0, Y_0, Y_1 are supposed to be real.

Definition 1 (Linear Independence)

Two functions y_1 and y_2 are said to be linearly independent on an interval I if neither of them is a constant multiple of the other on I, the functions are said to be linearly dependent otherwise.

Theorem 2 (Representation of Solutions)

If y_1 and y_2 are two linearly independent solutions of the differential equation

$$a\,y'' + b\,y' + c\,y = 0$$

then the constants c_1 and c_2 can be chosen such that

$$y(t) = c_1\,y_1(t) + c_2\,y_2(t)$$

is a solution to the initial value problem

$$a\,y'' + b\,y' + c\,y = 0 \qquad y(t_0) = Y_0,\ y'(t_0) = Y_1$$

For arbitrary c_1 and c_2, the expression $y(t) = c_1\,y_1(t) + c_2\,y_2(t)$ is known as the **general solution** of the differential equation $a\,y'' + b\,y' + c\,y = 0$. Technically, the general solution is a two-parameter family of functions.

Lemma 1

If y_1 and y_2 denote two solutions to the differential equation $a\,y'' + b\,y' + c\,y = 0$ and there exists a number τ for which

$$y_1(\tau)\,y_2'(\tau) - y_1'(\tau)\,y_2(\tau) = 0$$

then the functions y_1 and y_2 are linearly dependent on $(-\infty, \infty)$.

Definition 2 (the Wronskian determinant)

The Wronskian determinant for two functions y_1 and y_2 is given by

$$W(t) = y_1(t)\,y_2'(t) - y_1'(t)\,y_2(t)$$

Lemma 2

If the functions y_1 and y_2 are linearly dependent on the interval I, then the corresponding Wronskian determinant is identically equal to zero on I.

Theorem 3 (Linear Independence of Solutions)

Two solutions y_1 and y_2 of the differential equation $a\,y'' + b\,y' + c\,y = 0$ are linearly independent if and only if the corresponding Wronskian determinant is never zero on the interval $(-\infty, \infty)$.

Examples

Initializations

```
> restart;
  with(linalg):
Warning, the protected names norm and trace have been redefined and unprotected
```

Example 13.1

Consider the initial value problem

$$2\,y''+3\,y'-5\,y=0,\ y(0)=3,\ y'(0)=-2$$

1) Find two solutions Y_1 and Y_2 of the differential equation.
2) Use the Wronskian to verify that your choice of Y_1 and Y_2 forms a linearly independent pair.
3) Assemble the general solution of the differential equation.
4) Implement the initial conditions and solve the initial value problem.

Part 1

Find and solve the auxiliary equation.

```
> deq:=2*diff(y(x), x$2)+3*diff(y(x), x)-5*y(x)=0;
```

$$deq := 2\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+3\left(\frac{d}{dx}\mathrm{y}(x)\right)-5\,\mathrm{y}(x)=0$$

```
> aux_eq:=factor(eval(subs(y(x)=exp(r*x), deq))/exp(r*x));
```

$$aux\_eq := (2\,r+5)\,(r-1)=0$$

The eigenvalues are given by

```
> evals:=solve(aux_eq, r);
```

$$evals := 1, \frac{-5}{2}$$

Assemble the corresponding solutions.

```
> Y[1]:=exp(evals[1]*x);
  Y[2]:=exp(evals[2]*x);
```

$$Y_1 := \mathbf{e}^{x}$$

$$Y_2 := \mathbf{e}^{\left(-\frac{5\,x}{2}\right)}$$

```
>
```

Part 2

Compute the Wronskian determinant.

```
> W:=simplify(det(wronskian([Y[1], Y[2]], x)));
```

$$\mathrm{W} := -\frac{7}{2}\,\mathbf{e}^{\left(-\frac{3\,x}{2}\right)}$$

Clearly, W is never zero, so by Theorem 3, the solutions Y_1 and Y_2 are linearly independent.

Part 3

By Theorem 2, the general solution of the differential equation is an arbitrary linear combination of Y_1 and Y_2.

```
> gensol:=add(c[k]*Y[k], k=1..2);
```

$$gensol := c_1\,\mathbf{e}^{x}+c_2\,\mathbf{e}^{\left(-\frac{5\,x}{2}\right)}$$

```
>
```

Part 4

Implement the initial conditions and determine the constants c_1 and c_2.

```
> eq1:=eval(subs(x=0, gensol)=3);
  eq2:=eval(subs(x=0, diff(gensol, x))=-2);
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1+c_2=3$$

$$eq2 := c_1 - \frac{5}{2} c_2 = -2$$

$$val\_c := \{c_2 = \frac{10}{7}, c_1 = \frac{11}{7}\}$$

Assemble the solution of the initial value problem.

```
> sol:=subs(val_c, gensol);
```

$$sol := \frac{11}{7} e^{x} + \frac{10}{7} e^{\left(-\frac{5x}{2}\right)}$$

Example 13.2

The method used in Example 13.1 generalizes to higher order equations.

*) The functions $y_1, y_2, \ldots, y_n$ are said to be linearly independent on an interval I, if none of them can be written as a linear combination of the others on I.

**) The Wonskian determinant for the functions $y_1, y_2, \ldots, y_n$, is defined as

$$W = \begin{vmatrix} y_1 & y_2 & \cdots & y_n \\ y_1' & y_2' & \cdots & y_n' \\ \vdots & \vdots & & \vdots \\ y_1^{(n-1)} & y_2^{(n-1)} & \cdots & y_n^{(n-1)} \end{vmatrix}$$

***) Lemma 2 and Theorem 3 generalize to n-th order, linear, homogeneous equations with constant coefficients, and solutions $y_1, y_2, \ldots, y_n$ thereof.

Consider the initial value problem.

$$y''' - 4y'' + 14y' - 20y = 0, \; y(0) = 5, \; y'(0) = 2, \; y''(0) = -3$$

1) Find three solutions Y_1, Y_2 and Y_3 of the differential equation.
2) Use the Wronskian to verify that your choice of Y_1, Y_2 and Y_3 forms a linearly independent triple.
3) Assemble the general solution of the differential equation.
4) Implement the initial conditions and solve the initial value problem.

Part 1

Find and solve the auxiliary equation.

```
> deq:=diff(y(x), x$3)-4*diff(y(x), x$2)+14*diff(y(x), x)-20*y(x)=0;
```

$$deq := \left(\frac{d^3}{dx^3} y(x)\right) - 4\left(\frac{d^2}{dx^2} y(x)\right) + 14\left(\frac{d}{dx} y(x)\right) - 20\, y(x) = 0$$

```
> aux_eq:=factor(eval(subs(y(x)=exp(r*x), deq))/exp(r*x));
```

$$aux\_eq := (r-2)(r^2 - 2r + 10) = 0$$

```
> evals:=solve(aux_eq, r);
```

$$evals := 2, 1 + 3I, 1 - 3I$$

Assemble the corresponding solutions.

```
> Y[1]:=exp(evals[1]*x);
  Y[2]:=exp(Re(evals[2])*x)*cos(Im(evals[2])*x);
  Y[3]:=exp(Re(evals[2])*x)*sin(Im(evals[2])*x);
```

$$Y_1 := e^{(2x)}$$

$$Y_2 := e^{x} \cos(3x)$$

$$Y_3 := e^{x} \sin(3x)$$

```
>
```

Part 2

Compute the Wronskian determinant.

```
> W:=simplify(det(wronskian([seq(Y[k], k=1..3)], x)));
```

$$W := 30\, e^{(4x)}$$

Part 3

The general solution of the differential equation is an arbitrary linear combination of Y_1, Y_2 and Y_3.

```
> gensol:=add(c[k]*Y[k], k=1..3);
```

$$gensol := c_1\,\mathbf{e}^{(2\,x)} + c_2\,\mathbf{e}^x \cos(3\,x) + c_3\,\mathbf{e}^x \sin(3\,x)$$

```
>
```

Part 4

Implement the initial conditions and determine the constants c_1, c_2 and c_3.

```
> eq1:=eval(subs(x=0, gensol)=5);
  eq2:=eval(subs(x=0, diff(gensol, x))=2);
  eq3:=eval(subs(x=0, diff(gensol, x$2))=-3);
  val_c:=solve({seq(eq||k, k=1..3)}, {seq(c[k], k=1..3)});
```

$$eq1 := c_1 + c_2 = 5$$

$$eq2 := 2\,c_1 + c_2 + 3\,c_3 = 2$$

$$eq3 := 4\,c_1 - 8\,c_2 + 6\,c_3 = -3$$

$$val\_c := \{c_3 = \frac{-73}{30},\, c_2 = \frac{7}{10},\, c_1 = \frac{43}{10}\}$$

Assemble the solution of the initial value problem.

```
> sol:=subs(val_c, gensol);
```

$$sol := \frac{43}{10}\,\mathbf{e}^{(2\,x)} + \frac{7}{10}\,\mathbf{e}^x \cos(3\,x) - \frac{73}{30}\,\mathbf{e}^x \sin(3\,x)$$

```
>
```